

gost705 и gost705s — стили оформления библиографии в соответствии с ГОСТ Р 7.0.5-2008

Станислав Кручинин

E-mail: stanislav.kruchinin@gmail.com

Версия 1.1.5

2009/03/23

Содержание

1	Исходный текст	1
1.1	Поля, используемые в bib-файлах	1
1.2	Объявление и инициализация переменных и констант	2
1.3	Переключение языка	2
1.4	Вывод библиографической записи	2
1.5	Вывод блоков и ограничителей	3
1.6	Общепринятые сокращения	6
1.7	Формат элементов библиографических ссылок	11
1.8	Описание типов библиографических ссылок	20
1.9	Сокращения названий месяцев	26
1.10	Сокращения названий журналов	26
1.10.1	Физика и астрономия	26
1.10.2	Оптика	27
1.10.3	Физика конденсированного состояния	28
1.11	Функции для сортировки списка литературы по именам авторов и названиям	28
1.12	Вычисление длины табуляции между номером и текстом	30
1.13	Формирование окружения thebibliography	30
1.13.1	Преамбула	30
1.13.2	Начало окружения	31
1.13.3	Конец окружения	31
2	Предметный указатель	32

1 Исходный текст

```
1
2 <*gost705 | gost705s>
3
```

1.1 Поля, используемые в bib-файлах

```
4 ENTRY
5 { address
6   annote
7   author
8   booktitle
9   bookauthor
10  chapter
11  edition
12  editor
13  compiler
14  howpublished
15  institution
16  journal
```

```

17     key
18     month
19     note
20     number
21     organization
22     pages
23     numpages
24     publisher
25     school
26     series
27     title
28     type
29     volume
30     year
31     language
32     booklanguage
33     url
34     urldate
35     isbn
36     doi
37 }
38 {}
39 { label }
40

```

1.2 Объявление и инициализация переменных и констант

```

41 INTEGERS { output.state before.all mid.sentence after.sentence after.block
42 after.dblslash after.slash after.column after.semicolumn }
43

```

init.state.consts

```

44 FUNCTION {init.state.consts}
45 {
46   #0 'before.all :=
47   #1 'mid.sentence :=
48   #2 'after.sentence :=
49   #3 'after.block :=
50   #4 'after.dblslash :=
51   #5 'after.slash :=
52   #6 'after.column :=
53   #7 'after.semicolumn :=
54 }
55
56 STRINGS { s t }
57
58 STRINGS { curlanguage }
59

```

1.3 Переключение языка

change.language

```

60 FUNCTION {change.language}
61 {
62   booklanguage empty$
63   { "" }
64   { booklanguage 'curlanguage :=
65     "\selectlanguageifdefined{"
66     curlanguage *
67     "}" *
68   }
69   if$
70 }
71

```

1.4 Вывод библиографической записи

output.nonnull

```

72 FUNCTION {output.nonnull}
73 {
74   swap$
75   output.state mid.sentence =
76   { " " * write$ }
77   { output.state after.block =
78     { add.period$ " " * write$
79       newline$
80       "\newblock " write$
81     }
82     { output.state before.all =
83       'write$
84       { output.state after.dblslash =
85         { "~//" * change.language * " " * write$ }
86         { output.state after.slash =
87           { "~/" * write$ }
88           { output.state after.column =
89             { ":" * write$ }
90             { output.state after.semicolumn =
91               { ";" * write$ }
92               { add.period$ " " * write$ }
93             if$
94           }
95           if$
96         }
97         if$
98       }
99       if$
100     }
101     if$
102   }
103   if$
104   mid.sentence 'output.state :=
105 }
106 if$
107 }
108
output
109 FUNCTION {output}
110 {
111   duplicate$ empty$
112   'pop$
113   'output.nonnull
114   if$
115 }
116
output.check
117 FUNCTION {output.check}
118 {
119   't :=
120   duplicate$ empty$
121   { pop$ "empty " t * " in " * cite$ * warning$ }
122   'output.nonnull
123   if$
124 }
125

```

1.5 Вывод блоков и ограничителей

fin.entry

```

126 FUNCTION {fin.entry}
127 {
128   add.period$
129   write$

```

```

130  newline$
131 }
132
new.block
133 FUNCTION {new.block}
134 {
135   output.state before.all =
136   'skip$
137   { after.block 'output.state := }
138   if$
139 }
140
new.dblslash
141 FUNCTION {new.dblslash}
142 {
143   output.state before.all =
144   'skip$
145   { after.dblslash 'output.state := }
146   if$
147 }
148
new.slash
149 FUNCTION {new.slash}
150 {
151   output.state before.all =
152   'skip$
153   { after.slash 'output.state := }
154   if$
155 }
156
new.column
157 FUNCTION {new.column}
158 {
159   output.state before.all =
160   'skip$
161   { after.column 'output.state := }
162   if$
163 }
164
new.semicolumn
165 FUNCTION {new.semicolumn}
166 {
167   output.state before.all =
168   'skip$
169   { after.semicolumn 'output.state := }
170   if$
171 }
172
new.sentence
173 FUNCTION {new.sentence}
174 {
175   output.state after.block =
176   'skip$
177   { output.state before.all =
178     'skip$
179     { after.sentence 'output.state := }
180     if$
181   }
182   if$
183 }
184

```

```

add.blank
185 FUNCTION {add.blank} { " " * before.all 'output.state := }
186

not
187 FUNCTION {not}
188 {
189   { #0 }
190   { #1 }
191   if$
192 }
193

and
194 FUNCTION {and}
195 {
196   'skip$
197   { pop$ #0 }
198   if$
199 }
200

or
201 FUNCTION {or}
202 {
203   { pop$ #1 }
204   'skip$
205   if$
206 }
207

non.stop
208 FUNCTION {non.stop}
209 {
210   duplicate$
211   "]" * add.period$
212   #-1 #1 substring$ "." =
213 }
214

new.block.checka
215 FUNCTION {new.block.checka}
216 {
217   empty$
218   'skip$
219   'new.block
220   if$
221 }
222

new.block.checkb
223 FUNCTION {new.block.checkb}
224 {
225   empty$
226   swap$ empty$
227   and
228   'skip$
229   'new.block
230   if$
231 }
232

new.sentence.checka
233 FUNCTION {new.sentence.checka}
234 {
235   empty$
236   'skip$
237   'new.sentence
238   if$

```

```

239 }
240
new.sentence.checkb
241 FUNCTION {new.sentence.checkb}
242 {
243   empty$
244   swap$ empty$
245   and
246   'skip$
247   'new.sentence
248   if$
249 }
250
field.or.null
251 FUNCTION {field.or.null}
252 {
253   duplicate$ empty$
254   { pop$ "" }
255   'skip$
256   if$
257 }
258
emphasize
259 FUNCTION {emphasize}
260 {
261   duplicate$ empty$
262   { pop$ "" }
263   { "\BibEmph{" swap$ * "}" * }
264   if$
265 }
266
space.word
267 FUNCTION {space.word} { " " swap$ * " " * }
268

```

1.6 Общепринятые сокращения

```

269 FUNCTION {bbl.edby} % { "\bbledby{" }
270 {
271   curlanguage "english" =
272   { "Ed. by " }
273   { curlanguage "russian" =
274     { "{\cyr\CYRP\cyro\cyrd\ \cyrr\cyre\cyrd.\ }" }
275     { "language is not defined: " language "edited by" * *
276       warning$ "edited by" }
277     if$
278   }
279   if$
280 }
281
bbl.cmplr
282 FUNCTION {bbl.cmplr}
283 {
284   curlanguage "english" =
285   { "Compiler " }
286   { curlanguage "russian" =
287     { "{\cyr\CYRS\cyro\cyrs\cyrt.\ }" }
288     { "language is not defined: " language "compiler" * * warning$ "compiler" }
289     if$
290   }
291   if$
292 }
293

```

```

bbl.edition
294 FUNCTION {bbl.edition} % { "\bbledition{" }
295 {
296   curlanguage "english" =
297     { "edition" }
298   { curlanguage "russian" =
299     { "{\cyr\cyri\cyrz\cyrd.}" }
300     { "language is not defined: " language "edition" * * warning$ "edition" }
301     if$
302   }
303   if$
304 }
305

bbl.vvolume
306 FUNCTION {bbl.vvolume} % { "\bblVolume{" }
307 {
308   curlanguage "english" =
309     { "Volume" }
310   { curlanguage "russian" =
311     { "{\cyr\CYRT\cyro\cyrm}" }
312     { "language is not defined: " language "Volume" * * warning$ "Volume" }
313     if$
314   }
315   if$
316 }
317

bbl.vvol
318 FUNCTION {bbl.vvol} % { "\bblVol{" }
319 {
320   curlanguage "english" =
321     { "Vol." }
322   { curlanguage "russian" =
323     { "{\cyr\CYRT.}" }
324     { "language is not defined: " language "Vol." * * warning$ "Vol." }
325     if$
326   }
327   if$
328 }
329

bbl.iissue
330 FUNCTION {bbl.iissue} % { "\bblIssue{" }
331 {
332   curlanguage "english" =
333     { "Issue" }
334   { curlanguage "russian" =
335     { "{\cyr\CYRV\cyrery\cyrp\cyru\cyrs\cyrk}" }
336     { "language is not defined: " language * warning$ "Issue" }
337     if$
338   }
339   if$
340 }
341

bbl.iiss
342 FUNCTION {bbl.iiss} % { "\bblIss{" }
343 {
344   curlanguage "english" =
345     { "Iss." }
346   { curlanguage "russian" =
347     { "{\cyr\CYRV\cyrery\cyrp.}" }
348     { "language is not defined: " language * warning$ "Issue" }
349     if$
350   }
351   if$
352 }
353

```

```

bbl.of
354 FUNCTION {bbl.of} % { "\bbl of{}" }
355 {
356   curlanguage "english" =
357     { "of" }
358   { curlanguage "russian" =
359     { "{\cyr\cyri\cyrr}" }
360     { "language is not defined: " language "of" * * warning$ "of" }
361     if$
362   }
363   if$
364 }
365

bbl.etal
366 FUNCTION {bbl.etal} % { "\bbl of{}" }
367 {
368   curlanguage "english" =
369     { "et~al." }
370   { curlanguage "russian" =
371     { "{\cyr\cyri~\cyrd\cyrr.}" }
372     { "language is not defined: " language "at al" * * warning$ "at al." }
373     if$
374   }
375   if$
376 }
377

bbl.nnumber
378 FUNCTION {bbl.nnumber} % { "\bblNumber{}" }
379 {
380   curlanguage "english" =
381     { "Number" }
382   { curlanguage "russian" =
383     { "{\cyr\CYRN\cyro\cyrm\cyre\cyrr}" }
384     { "language is not defined: " language "Number" * * warning$ "Number" }
385     if$
386   }
387   if$
388 }
389

bbl.number
390 FUNCTION {bbl.number} % { "\bblnumber{}" }
391 {
392   curlanguage "english" =
393     { "number" }
394   { curlanguage "russian" =
395     { "{\cyr\cyrn\cyro\cyrm\cyre\cyrr}" }
396     { "language is not defined: " language "number" * * warning$ "number" }
397     if$
398   }
399   if$
400 }
401

bbl.nr
402 FUNCTION {bbl.nr} % { "\bblno{}" }
403 {
404   curlanguage "english" =
405     { "no." }
406   { curlanguage "russian" =
407     { "{\cyr\textnumero}" }
408     { "language is not defined: " language "no." * * warning$ "no." }
409     if$
410   }
411   if$
412 }
413

```

```

bbl.nnr
414 FUNCTION {bbl.nnr} % { "\bblno{" }
415 {
416   curlanguage "english" =
417   { "No." }
418   { curlanguage "russian" =
419     { "{\cyr\textnumero}" }
420     { "language is not defined: " language "No." * * warning$ "No." }
421     if$
422   }
423   if$
424 }
425

bbl.in
426 FUNCTION {bbl.in} % { "\bblin{" }
427 {
428   curlanguage "english" =
429   { "in" }
430   { curlanguage "russian" =
431     { "{\cyr\cyrv}" }
432     { "language is not defined: " language "in" * * warning$ "in" }
433     if$
434   }
435   if$
436 }
437

bbl.iin
438 FUNCTION {bbl.iin} % { "\bblIn{" }
439 {
440   curlanguage "english" =
441   { "In" }
442   { curlanguage "russian" =
443     { "{\cyr\CYRV}" }
444     { "language is not defined: " language "In" * * warning$ "In" }
445     if$
446   }
447   if$
448 }
449

bbl.pages
450 FUNCTION {bbl.pages} % { "\bblpp." }
451 {
452   curlanguage "english" =
453   { "pp." }
454   { curlanguage "russian" =
455     { "{\cyr\cyrs.}" }
456     { "language is not defined: " language "pp." * * warning$ "pp." }
457     if$
458   }
459   if$
460 }
461

bbl.page
462 FUNCTION {bbl.page} % { "\bblp." }
463 {
464   curlanguage "english" =
465   { "p." }
466   { curlanguage "russian" =
467     { "{\cyr\cyrs.}" }
468     { "language is not defined: " language "p." * * warning$ "p." }
469     if$
470   }
471   if$
472 }
473

```

```

bbl.ppages
474 FUNCTION {bbl.ppages} % { "\bblPp." }
475 {
476   curlanguage "english" =
477     { "Pp." }
478   { curlanguage "russian" =
479     { "{\cyr\CYRS.}" }
480     { "language is not defined: " language "Pp." * * warning$ "Pp." }
481     if$
482   }
483   if$
484 }
485

bbl.ppage
486 FUNCTION {bbl.ppage} % { "\bblP." }
487 {
488   curlanguage "english" =
489     { "P." }
490   { curlanguage "russian" =
491     { "{\cyr\CYRS.}" }
492     { "language is not defined: " language "P." * * warning$ "P." }
493     if$
494   }
495   if$
496 }
497

bbl.techrep
498 FUNCTION {bbl.techrep}
499 {
500   curlanguage "english" =
501     { "Tech. Rep." }
502   { curlanguage "russian" =
503     { "{\cyr\CYRT\cyre\cyrh.\ \cyrd\cyro\cyrk\cyrl\cyra\cyrd.}" }
504     { "language is not defined: " language "Tech. Rep." * * warning$ "P." }
505     if$
506   }
507   if$
508 }
509

bbl.mthesis
510 FUNCTION {bbl.mthesis}
511 {
512   curlanguage "english" =
513     { "Master's thesis" }
514   { curlanguage "russian" =
515     { "{\cyr\CYRM\cyra\cyrg\cyri\cyrs\cyrt\cyre\cyrr\cyrs\cyrk\cyra\cyrya\ "
516       "\cyrd\cyri\cyrs\cyrs\cyre\cyrr\cyrt\cyra\cyrc\cyri\cyrya}" * }
517     { "language is not defined: " language "Master's thesis" * * warning$ "P." }
518     if$
519   }
520   if$
521 }
522

bbl.phdthesis
523 FUNCTION {bbl.phdthesis}
524 {
525   curlanguage "english" =
526     { "Ph.\,D. thesis" }
527   { curlanguage "russian" =
528     { "{\cyr\CYRK\cyra\cyrn\cyrd\cyri\cyrd\cyra\cyrt\cyrs\cyrk\cyra\cyrya\ "
529       "\cyrd\cyri\cyrs\cyrs\cyre\cyrr\cyrt\cyra\cyrc\cyri\cyrya}" * }
530     { "language is not defined: " language "Ph.\,D. thesis" * * warning$ "P." }
531     if$
532   }
533   if$

```

```

534 }
535
bbl.docdisser
536 FUNCTION {bbl.docdisser}
537 {
538   curlanguage "english" =
539     { "Dr.\,Sci. dissertation" }
540     { curlanguage "russian" =
541       { "\cyr\CYRD\cyro\cyrk\cyrt\cyro\cyrr\cyrs\cyrk\cyra\cyrya\ "
542         "\cyrd\cyri\cyrs\cyrs\cyre\cyrr\cyrt\cyra\cyrc\cyri\cyrya}" * }
543       { "language is not defined: " language "Dr.\,Sci. dissertation" * *
544         warning$ "P." }
545       if$
546     }
547   if$
548 }
549

```

1.7 Формат элементов библиографических ссылок

```

550 INTEGERS { nameptr namesleft numnames }
551
format.names
552 FUNCTION {format.names}
553 {
554   's :=
555   #1 'nameptr :=
556   s num.names$ 'numnames :=
557   numnames 'namesleft :=
558   { namesleft #0 > }
559   { s nameptr
560     "{vv~}{ll}{~f.}{, jj}" format.name$ 't :=
561     nameptr #1 >
562     { nameptr #4 =
563       numnames #4 > and
564       { "others" 't :=
565         #1 'namesleft := }
566       'skip$
567       if$
568       namesleft #1 >
569       { ", " * t * }
570       { t "others" =
571         t "~others" =
572         or
573         { " " * bbl.etal * }
574         { ", " * t * }
575       if$
576     }
577     if$
578   }
579   't
580   if$
581   nameptr #1 + 'nameptr :=
582   namesleft #1 - 'namesleft :=
583 }
584 while$
585 }
586
format.names.rev
587 FUNCTION {format.names.rev}
588 {
589   's :=
590   #1 'nameptr :=
591   s num.names$ 'numnames :=
592   numnames 'namesleft :=
593   { namesleft #0 > }

```

```

594 { s nameptr
595 "{f.}{~vv}{~ll}{, jj}" format.name$ 't :=
596 nameptr #1 >
597 { nameptr #4 =
598   numnames #4 > and
599   { "others" 't :=
600     #1 'namesleft := }
601   'skip$
602   if$
603   namesleft #1 >
604   { ", " * t * }
605   { t "others" =
606     t "~others" =
607     or
608     { " " * bbl.etal * }
609     { ", " * t * }
610   if$
611   }
612   if$
613   }
614   't
615   if$
616   nameptr #1 + 'nameptr :=
617   namesleft #1 - 'namesleft :=
618   }
619 while$
620 }
621

```

format.authors

```

622 FUNCTION {format.authors}
623 {
624   author empty$
625   { "" }
626   { author format.names emphasize }
627   if$
628 }
629

```

format.bookauthors

```

630 FUNCTION {format.bookauthors}
631 {
632   bookauthor empty$
633   { "" }
634   { bookauthor format.names }
635   if$
636 }
637

```

format.authors.after

```

638 FUNCTION {format.authors.after}
639 {
640   author empty$
641   { "" }
642   { author format.names.rev }
643   if$
644 }
645

```

format.bookauthors.after

```

646 FUNCTION {format.bookauthors.after}
647 {
648   bookauthor empty$
649   { "" }
650   { bookauthor format.names.rev }
651   if$
652 }
653

```

```

format.editors.after
654 FUNCTION {format.editors.after}
655 {
656   editor empty$
657   { "" }
658   { bbl.edby editor format.names.rev * }
659   if$
660 }
661

format.compiler.after
662 FUNCTION {format.compiler.after}
663 {
664   compiler empty$
665   { "" }
666   { bbl.cmplr compiler format.names.rev * }
667   if$
668 }
669

format.title
670 FUNCTION {format.title}
671 {
672   title empty$
673   { "" }
674   { title }
675   if$
676 }
677

format.date
678 FUNCTION {format.date}
679 {
680   year empty$
681   { month empty$
682     { "" }
683     { "there's a month but no year in " cite$ * warning$ month }
684     if$
685   }
686   { month empty$
687     'year
688     { year ". " quote$ "--- " month * * * * }
689     if$
690   }
691   if$
692 }
693

output.address.publisher.date
694 FUNCTION {output.address.publisher.date}
695 {
696   address empty$
697   'skip$
698   { address output
699     publisher empty$
700     'skip$
701     { new.column }
702     if$
703   }
704   if$
705   publisher output
706   format.date output
707 }
708

output.bibitem
709 FUNCTION {output.bibitem}
710 {
711   newline$

```

```

712 "\bibitem[" write$
713 "\BibPrefix" write$
714 label write$
715 "\BibPostfix" write$
716 "]" write$
717 cite$ write$
718 "]" write$
719 newline$
720 language empty$
721 { "english" 'curlanguage := }
722 { language 'curlanguage := }
723 if$
724 "\selectlanguageifdefined{" write$
725 curlanguage write$
726 "}" write$ newline$
727 ""
728 before.all 'output.state :=
729 }
730
n.dashify
731 FUNCTION {n.dashify}
732 {
733 't :=
734 ""
735 { t empty$ not }
736 { t #1 #1 substring$ "-" =
737   { t #1 #2 substring$ "--" = not
738     { "--" *
739       t #2 global.max$ substring$ 't :=
740     }
741     { { t #1 #1 substring$ "-" = }
742       { "-" *
743         t #2 global.max$ substring$ 't :=
744       }
745       while$
746     }
747     if$
748   }
749   { t #1 #1 substring$ *
750     t #2 global.max$ substring$ 't :=
751   }
752   if$
753 }
754 while$
755 }
756
word.in
757 FUNCTION {word.in} { bbl.iin " " * }
758
format.btitle
759 FUNCTION {format.btitle} { title }
760
tie.or.space.connect
761 FUNCTION {tie.or.space.connect}
762 {
763 duplicate$ text.length$ #3 <
764 { "~" }
765 { " " }
766 if$
767 swap$ * *
768 }
769
tie.connect
770 FUNCTION {tie.connect} { "~" swap$ * * }

```

```

771
either.or.check
772 FUNCTION {either.or.check}
773 {
774   empty$
775   'pop$
776   { "can't use both " swap$ * " fields in " * cite$ * warning$ }
777   if$
778 }
779

format.bvolume
780 FUNCTION {format.bvolume}
781 {
782   volume empty$
783   { "" }
784   { bbl.vvol volume tie.connect
785     series empty$
786     'skip$
787     { bbl.of space.word * series emphasize * }
788     if$
789     "volume and number" number either.or.check
790   }
791   if$
792 }
793

format.number.series
794 FUNCTION {format.number.series}
795 {
796   volume empty$
797   { number empty$
798     { series field.or.null }
799     { series empty$
800       { "there's a number but no series in " cite$ * warning$
801         bbl.nnr
802       }
803       { new.sentence
804         series
805         bbl.nr
806         tie.or.space.connect
807       }
808       if$
809       number tie.or.space.connect
810     }
811     if$
812   }
813   { "" }
814   if$
815 }
816

is.num
817 FUNCTION {is.num}
818 {
819   chr.to.int$
820   duplicate$ "0" chr.to.int$ < not
821   swap$ "9" chr.to.int$ > not and
822 }
823

extract.num
824 FUNCTION {extract.num}
825 {
826   duplicate$ 't :=
827   "" 's :=
828   { t empty$ not }
829   { t #1 #1 substring$

```

```

830     t #2 global.max$ substring$ 't :=
831     duplicate$ is.num
832     { s swap$ * 's := }
833     { pop$ "" 't := }
834     if$
835   }
836   while$
837   s empty$
838   'skip$
839   { pop$ s }
840   if$
841 }
842

convert.edition
843 FUNCTION {convert.edition} { edition }
844

format.edition
845 FUNCTION {format.edition}
846 {
847   edition empty$
848   { "" }
849   { output.state mid.sentence =
850     { convert.edition "l" change.case$ " " * bbl.edition * }
851     { convert.edition "t" change.case$ " " * bbl.edition * }
852     if$
853   }
854   if$
855 }
856
857 INTEGERS { multiresult }
858

multi.page.check
859 FUNCTION {multi.page.check}
860 {
861   't :=
862   #0 'multiresult :=
863   { multiresult not
864     t empty$ not
865     and
866   }
867   { t #1 #1 substring$
868     duplicate$ "-" =
869     swap$ duplicate$ "," =
870     swap$ "+" =
871     or or
872     { #1 'multiresult := }
873     { t #2 global.max$ substring$ 't := }
874     if$
875   }
876   while$
877   multiresult
878 }
879

format.pages
880 FUNCTION {format.pages}
881 {
882   pages empty$
883   { "" }
884   { pages multi.page.check
885     { bbl.ppages pages n.dashify tie.connect }
886     { bbl.ppage pages tie.connect }
887     if$
888   }
889   if$
890 }

```

```

891
format.pages.page
892 FUNCTION {format.pages.page}
893 {
894   pages empty$
895     { numpages empty$
896       { "" }
897       { numpages bbl.pages tie.connect }
898       if$
899     }
900     { format.pages }
901     if$
902 }
903
format.vol.num.pages
904 FUNCTION {format.vol.num.pages}
905 {
906   volume field.or.null
907   number empty$
908     'skip$
909     { ", no." number tie.or.space.connect *
910       volume empty$
911         { "there's a number but no volume in " cite$ * warning$ }
912         'skip$
913         if$
914       }
915     if$
916   pages empty$
917     'skip$
918     { duplicate$ empty$
919       { pop$ format.pages }
920       { ": " * pages n.dashify * }
921       if$
922     }
923   if$
924 }
925
format.volume
926 FUNCTION {format.volume}
927 {
928   volume empty$
929     { "" }
930     { bbl.vvol volume tie.or.space.connect }
931     if$
932 }
933
format.number
934 FUNCTION {format.number}
935 {
936   number empty$
937     { "" }
938     { bbl.nr number tie.or.space.connect }
939     if$
940 }
941
empty.misc.check
942 FUNCTION {empty.misc.check}
943 {
944   author empty$ title empty$ howpublished empty$
945   month empty$ year empty$ note empty$
946   and and and and and
947   key empty$ not and
948   { "all relevant fields are empty in " cite$ * warning$ }
949   'skip$

```

```

950   if$
951 }
952
format.thesis.type
953 FUNCTION {format.thesis.type}
954 {
955   type empty$
956   'skip$
957   { pop$
958     type "t" change.case$
959   }
960   if$
961 }
962
format.tr.number
963 FUNCTION {format.tr.number}
964 {
965   type empty$
966   { bbl.techrep }
967   'type
968   if$
969   number empty$
970   { "t" change.case$ }
971   { number tie.or.space.connect }
972   if$
973 }
974
author.before
975 FUNCTION {author.before}
976 {
977   author empty$
978   'skip$
979   { format.authors output
980     new.sentence
981   }
982   if$
983 }
984
bookauthor.before
985 FUNCTION {bookauthor.before}
986 {
987   bookauthor empty$
988   'skip$
989   { format.bookauthors output
990     new.sentence
991   }
992   if$
993 }
994
author.after
995 FUNCTION {author.after}
996 {
997   author empty$
998   'skip$
999   { author num.names$ #3 >
1000     { format.authors.after output
1001       new.semicolumn
1002     }
1003     'skip$
1004     if$
1005   }
1006   if$
1007 }
1008

```

```

bookauthor.after
1009 FUNCTION {bookauthor.after}
1010 {
1011   bookauthor empty$
1012   'skip$
1013   { bookauthor num.names$ #3 >
1014     { format.bookauthors.after output
1015       new.semicolumn
1016     }
1017     'skip$
1018     if$
1019   }
1020   if$
1021 }
1022

editor.organization.after
1023 FUNCTION {editor.organization.after}
1024 {
1025   compiler empty$
1026   'skip$
1027   { format.compiler.after output
1028     new.semicolumn
1029   }
1030   if$
1031   editor empty$
1032   'skip$
1033   { format.editors.after output
1034     new.semicolumn
1035   }
1036   if$
1037   organization empty$
1038   'skip$
1039   { organization output
1040     new.semicolumn
1041   }
1042   if$
1043 }
1044

format.url
1045 FUNCTION {format.url}
1046 {
1047   url empty$
1048   { "" }
1049   { "\BibUrl{" url * "}" * }
1050   if$
1051 }
1052

format.urldate
1053 FUNCTION {format.urldate}
1054 {
1055   urldate empty$
1056   { "" }
1057   { add.blank
1058     "\BibUrlDate{" urldate * "}" *
1059   }
1060   if$
1061 }
1062

output.url
1063 FUNCTION {output.url}
1064 {
1065   url empty$
1066   'skip$
1067   { format.url output
1068     format.urldate output

```

```

1069     }
1070     if$
1071 }
1072
format.annotate
1073 FUNCTION {format.annotate}
1074 {
1075     annotate empty$
1076     { "" }
1077     { after.sentence 'output.state :=
1078       "\BibAnnote{" annotate add.period$ * "}" *
1079     }
1080     if$
1081 }
1082
format.isbn
1083 FUNCTION {format.isbn}
1084 {
1085     isbn empty$
1086     { "" }
1087     { "ISBN:~\href{http://isbndb.com/search-all.html?kw=" isbn *
1088       "}{ " * isbn * "}" *
1089     }
1090     if$
1091 }
1092
add.doi
1093 FUNCTION {add.doi}
1094 { duplicate$ empty$
1095   'skip$
1096   { doi empty$
1097     'skip$
1098     { "\href{http://dx.doi.org/" doi * "}{ " * swap$ * "}" * }
1099     if$
1100   }
1101   if$
1102 }
1103

```

1.8 Описание типов библиографических ссылок

```

article
1104 FUNCTION {article}
1105 {
1106     output.bibitem
1107     author.before
1108     format.title "title" output.check
1109     new.dblslash
1110     journal emphasize add.doi "journal" output.check
1111     new.block
1112     format.date output
1113     new.block
1114     format.volume output
1115     format.number output
1116     new.block
1117     format.pages.page output
1118     new.block
1119     note output
1120     new.sentence
1121     output.url
1122     format.annotate output
1123     fin.entry
1124 }
1125
book

```

```

1126 FUNCTION {book}
1127 {
1128   output.bibitem
1129   author.before
1130   format.btitle add.doi "title" output.check
1131   editor.organization.after
1132   new.sentence
1133   format.number.series output
1134   new.block
1135   format.edition output
1136   new.block
1137   output.address.publisher.date
1138   new.block
1139   format.bvolume output
1140   new.block
1141   format.pages.page output
1142   new.block
1143   format.isbn output
1144   new.block
1145   note output
1146   new.sentence
1147   output.url
1148   format.annotate output
1149   fin.entry
1150 }
1151
booklet
1152 FUNCTION {booklet}
1153 {
1154   output.bibitem
1155   author.before
1156   format.title "title" output.check
1157   editor.organization.after
1158   new.block
1159   howpublished output
1160   address output
1161   format.date output
1162   new.block
1163   note output
1164   new.sentence
1165   output.url
1166   format.annotate output
1167   fin.entry
1168 }
1169
inbook
1170 FUNCTION {inbook}
1171 {
1172   output.bibitem
1173   author.before
1174   format.btitle add.doi "title" output.check
1175   new.dblslash
1176   bookauthor.before
1177   booktitle "booktitle" output.check
1178   editor.organization.after
1179   new.block
1180   format.edition output
1181   new.block
1182   format.number.series output
1183   new.sentence
1184   output.address.publisher.date
1185   new.block
1186   format.bvolume output
1187   new.block
1188   format.pages.page output

```

```

1189 new.block
1190 format.isbn output
1191 new.block
1192 note output
1193 new.sentence
1194 output.url
1195 format.annotate output
1196 fin.entry
1197 }
1198
incollection
1199 FUNCTION {incollection}
1200 {
1201   output.bibitem
1202   author.before
1203   new.sentence
1204   format.title add.doi "title" output.check
1205   new.dblslash
1206   booktitle "booktitle" output.check
1207   new.slash
1208   editor.organization.after
1209   new.block
1210   output.address.publisher.date
1211   new.block
1212   format.bvolume output
1213   format.number.series output
1214   new.block
1215   format.pages.page output
1216   new.block
1217   note output
1218   new.block
1219   output.url
1220   format.annotate output
1221   fin.entry
1222 }
1223
inproceedings
1224 FUNCTION {inproceedings}
1225 {
1226   output.bibitem
1227   author.before
1228   new.sentence
1229   format.title add.doi "title" output.check
1230   new.dblslash
1231   booktitle "booktitle" output.check
1232   new.slash
1233   editor.organization.after
1234   new.block
1235   format.bvolume output
1236   format.number.series output
1237   new.block
1238   address empty$
1239     { publisher output
1240       format.date "year" output.check
1241     }
1242     { address output.nonnull
1243       new.column
1244       publisher output
1245       format.date "year" output.check
1246     }
1247   if$
1248   new.block
1249   format.pages.page output
1250   new.block
1251   note output

```

```

1252 new.sentence
1253 output.url
1254 format.annotate output
1255 fin.entry
1256 }
1257
conference
1258 FUNCTION {conference} { inproceedings }
1259
manual
1260 FUNCTION {manual}
1261 {
1262   output.bibitem
1263   author empty$
1264   { organization empty$
1265     'skip$
1266     { organization output.nonnull
1267       address output
1268     }
1269     if$
1270   }
1271   { format.authors output.nonnull }
1272   if$
1273   new.block
1274   format.btitle "title" output.check
1275   author empty$
1276   { organization empty$
1277     { address new.block.checka
1278       address output
1279     }
1280     'skip$
1281     if$
1282   }
1283   { organization address new.block.checkb
1284     organization output
1285     address output
1286   }
1287   if$
1288   format.edition output
1289   format.date output
1290   new.block
1291   note output
1292   new.sentence
1293   output.url
1294   format.annotate output
1295   fin.entry
1296 }
1297
mastersthesis
1298 FUNCTION {mastersthesis}
1299 {
1300   output.bibitem
1301   format.authors "author" output.check
1302   new.block
1303   format.btitle add.doi "title" output.check
1304   new.block
1305   bbl.mthesis format.thesis.type output.nonnull
1306   school "school" output.check
1307   address output
1308   format.date "year" output.check
1309   new.block
1310   note output
1311   new.block
1312   output.url

```

```

1313     format.annotate output
1314     fin.entry
1315 }
1316
misc
1317 FUNCTION {misc}
1318 {
1319     output.bibitem
1320     format.authors "author" output.check
1321     new.block
1322     format.title "title" output.check
1323     new.block
1324     howpublished output
1325     new.block
1326     format.date output
1327     new.block
1328     note output
1329     new.sentence
1330     output.url
1331     format.annotate output
1332     fin.entry
1333 }
1334
phdthesis
1335 FUNCTION {phdthesis}
1336 {
1337     output.bibitem
1338     format.authors "author" output.check
1339     new.sentence
1340     format.btitle add.doi "title" output.check
1341     new.column
1342     bbl.phdthesis format.thesis.type output.nonnull
1343     new.slash
1344     school "school" output.check
1345     new.block
1346     output.address.publisher.date
1347     new.block
1348     format.pages.page output
1349     note output
1350     new.sentence
1351     output.url
1352     format.annotate output
1353     fin.entry
1354 }
1355
docdisser
1356 FUNCTION {docdisser}
1357 {
1358     output.bibitem
1359     format.authors "author" output.check
1360     new.sentence
1361     format.btitle add.doi "title" output.check
1362     new.column
1363     bbl.docdisser format.thesis.type output.nonnull
1364     new.slash
1365     school "school" output.check
1366     new.block
1367     output.address.publisher.date
1368     new.block
1369     format.pages.page output
1370     note output
1371     new.sentence
1372     output.url
1373     format.annotate output

```

```

1374   fin.entry
1375 }
1376
proceedings
1377 FUNCTION {proceedings}
1378 {
1379   output.bibitem
1380   format.btitle add.doi "title" output.check
1381   new.slash
1382   editor.organization.after
1383   new.block
1384   format.bvolume output
1385   format.number.series output
1386   address empty$
1387   { publisher output
1388     format.date "year" output.check
1389   }
1390   { address output.nonnull
1391     format.date "year" output.check
1392     new.sentence
1393     publisher output
1394   }
1395   if$
1396   new.block
1397   note output
1398   new.sentence
1399   output.url
1400   format.annotate output
1401   fin.entry
1402 }
1403
techreport
1404 FUNCTION {techreport}
1405 {
1406   output.bibitem
1407   author.before
1408   new.sentence
1409   format.title add.doi "title" output.check
1410   new.column
1411   format.tr.number output.nonnull
1412   editor.organization.after
1413   new.block
1414   address output
1415   new.column
1416   institution "institution" output.check
1417   format.date "year" output.check
1418   new.block
1419   note output
1420   new.sentence
1421   output.url
1422   format.annotate output
1423   fin.entry
1424 }
1425
unpublished
1426 FUNCTION {unpublished}
1427 {
1428   output.bibitem
1429   author.before
1430   format.btitle "title" output.check
1431   editor.organization.after
1432   new.block
1433   note "note" output.check
1434   new.sentence

```

```

1435 output.url
1436 format.annotate output
1437 fin.entry
1438 }
1439
default.type
1440 FUNCTION {default.type} { misc }
1441

```

1.9 Сокращения названий месяцев

```

1442 MACRO {jan} {"\bbljan{}}
1443 MACRO {feb} {"\bblfeb{}}
1444 MACRO {mar} {"\bblmar{}}
1445 MACRO {apr} {"\bblapr{}}
1446 MACRO {may} {"\bblmay{}}
1447 MACRO {jun} {"\bbljun{}}
1448 MACRO {jul} {"\bbljul{}}
1449 MACRO {aug} {"\bblaug{}}
1450 MACRO {sep} {"\bblsep{}}
1451 MACRO {oct} {"\bbloct{}}
1452 MACRO {nov} {"\bblnov{}}
1453 MACRO {dec} {"\bbldec{}}
1454

```

1.10 Сокращения названий журналов

1.10.1 Физика и астрономия

Код данной секции создан на основе файла `physjour.mbs` пакета `custom-bib`.

```

1455 MACRO {aa} {"Astron. \& Astrophys."}
1456 MACRO {aasup} {"Astron. \& Astrophys. Suppl. Ser."}
1457 MACRO {aj} {"Astron. J."}
1458 MACRO {aph} {"Acta Phys."}
1459 MACRO {advp} {"Adv. Phys."}
1460 MACRO {ajp} {"Amer. J. Phys."}
1461 MACRO {ajm} {"Amer. J. Math."}
1462 MACRO {amsci} {"Amer. Sci."}
1463 MACRO {anofd} {"Ann. Fluid Dyn."}
1464 MACRO {am} {"Ann. Math."}
1465 MACRO {ap} {"Ann. Phys. (NY)"}
1466 MACRO {adp} {"Ann. Phys. (Leipzig)"}
1467 MACRO {ao} {"Appl. Opt."}
1468 MACRO {apl} {"Appl. Phys. Lett."}
1469 MACRO {app} {"Astroparticle Phys."}
1470 MACRO {apj} {"Astrophys. J."}
1471 MACRO {apjsup} {"Astrophys. J. Suppl."}
1472 MACRO {apss} {"Astrophys. Space Sci."}
1473 MACRO {araa} {"Ann. Rev. Astron. Astrophys."}
1474 MACRO {baas} {"Bull. Amer. Astron. Soc."}
1475 MACRO {baps} {"Bull. Amer. Phys. Soc."}
1476 MACRO {cmp} {"Comm. Math. Phys."}
1477 MACRO {cpam} {"Commun. Pure Appl. Math."}
1478 MACRO {cppcf} {"Comm. Plasma Phys. \& Controlled Fusion"}
1479 MACRO {cpc} {"Comp. Phys. Comm."}
1480 MACRO {cvg} {"Class. Quant. Grav."}
1481 MACRO {cra} {"C. R. Acad. Sci. A"}
1482 MACRO {fed} {"Fusion Eng. \& Design"}
1483 MACRO {ft} {"Fusion Tech."}
1484 MACRO {grg} {"Gen. Relativ. Gravit."}
1485 MACRO {ieeens} {"IEEE Trans. Nucl. Sci."}
1486 MACRO {ieeeps} {"IEEE Trans. Plasma Sci."}
1487 MACRO {ijimw} {"Interntl. J. Infrared \& Millimeter Waves"}
1488 MACRO {ip} {"Infrared Phys."}
1489 MACRO {irp} {"Infrared Phys."}
1490 MACRO {jap} {"J. Appl. Phys."}

```

1491 MACRO {jasa} {"J. Acoust. Soc. America"}
1492 MACRO {jcp} {"J. Comp. Phys."}
1493 MACRO {jchp} {"J. Chem. Phys."}
1494 MACRO {jetp} {"Sov. Phys.--JETP"}
1495 MACRO {jfe} {"J. Fusion Energy"}
1496 MACRO {jfm} {"J. Fluid Mech."}
1497 MACRO {jmp} {"J. Math. Phys."}
1498 MACRO {jne} {"J. Nucl. Energy"}
1499 MACRO {jnec} {"J. Nucl. Energy, C: Plasma Phys., Accelerators, Thermonucl. Res."}
1500 MACRO {jnm} {"J. Nucl. Mat."}
1501 MACRO {jpc} {"J. Phys. Chem."}
1502 MACRO {jpp} {"J. Plasma Phys."}
1503 MACRO {jpsj} {"J. Phys. Soc. Japan"}
1504 MACRO {jsi} {"J. Sci. Instrum."}
1505 MACRO {jvst} {"J. Vac. Sci. & Tech."}
1506 MACRO {nat} {"Nature"}
1507 MACRO {nature} {"Nature"}
1508 MACRO {nedf} {"Nucl. Eng. & Design/Fusion"}
1509 MACRO {nf} {"Nucl. Fusion"}
1510 MACRO {nim} {"Nucl. Inst. & Meth."}
1511 MACRO {nimpr} {"Nucl. Inst. & Meth. in Phys. Res."}
1512 MACRO {np} {"Nucl. Phys."}
1513 MACRO {npb} {"Nucl. Phys. B"}
1514 MACRO {nt/f} {"Nucl. Tech./Fusion"}
1515 MACRO {npbpc} {"Nucl. Phys. B (Proc. Suppl.)"}
1516 MACRO {inc} {"Nuovo Cimento"}
1517 MACRO {nc} {"Nuovo Cimento"}
1518 MACRO {pf} {"Phys. Fluids"}
1519 MACRO {pfa} {"Phys. Fluids A: Fluid Dyn."}
1520 MACRO {pfb} {"Phys. Fluids B: Plasma Phys."}
1521 MACRO {pl} {"Phys. Lett."}
1522 MACRO {pla} {"Phys. Lett. A"}
1523 MACRO {plb} {"Phys. Lett. B"}
1524 MACRO {prep} {"Phys. Rep."}
1525 MACRO {pnas} {"Proc. Nat. Acad. Sci. USA"}
1526 MACRO {pp} {"Phys. Plasmas"}
1527 MACRO {ppcf} {"Plasma Phys. & Controlled Fusion"}
1528 MACRO {phitrs1} {"Philos. Trans. Roy. Soc. London"}
1529 MACRO {prl} {"Phys. Rev. Lett."}
1530 MACRO {pr} {"Phys. Rev."}
1531 MACRO {physrev} {"Phys. Rev."}
1532 MACRO {pra} {"Phys. Rev. A"}
1533 MACRO {prb} {"Phys. Rev. B"}
1534 MACRO {prc} {"Phys. Rev. C"}
1535 MACRO {prd} {"Phys. Rev. D"}
1536 MACRO {pre} {"Phys. Rev. E"}
1537 MACRO {ps} {"Phys. Scripta"}
1538 MACRO {procrs1} {"Proc. Roy. Soc. London"}
1539 MACRO {rmp} {"Rev. Mod. Phys."}
1540 MACRO {rsi} {"Rev. Sci. Inst."}
1541 MACRO {science} {"Science"}
1542 MACRO {sciam} {"Sci. Am."}
1543 MACRO {sam} {"Stud. Appl. Math."}
1544 MACRO {sjpp} {"Sov. J. Plasma Phys."}
1545 MACRO {spd} {"Sov. Phys.--Doklady"}
1546 MACRO {sptp} {"Sov. Phys.--Tech. Phys."}
1547 MACRO {spu} {"Sov. Phys.--Uspekhi"}
1548 MACRO {st} {"Sky and Telesc."}
1549

1.10.2 Оптика

Из файла photjour.mbs.

1550 MACRO {appopt} {"Appl. Opt."}
1551 MACRO {bell} {"Bell Syst. Tech. J."}
1552 MACRO {ell} {"Electron. Lett."}

```

1553 MACRO {jasp} {"J. Appl. Spectr."}
1554 MACRO {jqe} {"IEEE J. Quantum Electron."}
1555 MACRO {jlwt} {"J. Lightwave Technol."}
1556 MACRO {jmo} {"J. Mod. Opt."}
1557 MACRO {josa} {"J. Opt. Soc. America"}
1558 MACRO {josaa} {"J. Opt. Soc. Amer.~A"}
1559 MACRO {josab} {"J. Opt. Soc. Amer.~B"}
1560 MACRO {jdp} {"J. Phys. (Paris)"}
1561 MACRO {oc} {"Opt. Commun."}
1562 MACRO {ol} {"Opt. Lett."}
1563 MACRO {os} {"Opt. Spectrosc."}
1564 MACRO {phtl} {"IEEE Photon. Technol. Lett."}
1565 MACRO {pspie} {"Proc. Soc. Photo-Opt. Instrum. Eng."}
1566 MACRO {sjot} {"Sov. J. Opt. Technol."}
1567 MACRO {sjqe} {"Sov. J. Quantum Electron."}
1568 MACRO {sleb} {"Sov. Phys.--Leb. Inst. Rep."}
1569 MACRO {stph} {"Sov. Phys.--Techn. Phys."}
1570 MACRO {stphl} {"Sov. Techn. Phys. Lett."}
1571 MACRO {vr} {"Vision Res."}
1572 MACRO {zph} {"Z. f. Physik"}
1573 MACRO {zphb} {"Z. f. Physik~B"}
1574 MACRO {zphd} {"Z. f. Physik~D"}
1575

```

1.10.3 Физика конденсированного состояния

```

1576 MACRO {sse} {"Solid-State Electron."}
1577 MACRO {pss} {"Phys. Sol. State"}
1578 MACRO {sst} {"Semicond. Sci. Tech."}
1579 MACRO {nl} {"Nano Lett."}
1580
1581 READ
1582

```

1.11 Функции для сортировки списка литературы по именам авторов и названиям

```

1583 <*gost705s>

sortify
1584 FUNCTION {sortify}
1585 {
1586   purify$
1587   "1" change.case$
1588 }
1589
1590 INTEGERS { len }
1591

chop.word
1592 FUNCTION {chop.word}
1593 {
1594   's :=
1595   'len :=
1596   s #1 len substring$ =
1597   { s len #1 + global.max$ substring$ }
1598   's
1599   if$
1600 }
1601

sort.format.names
1602 FUNCTION {sort.format.names}
1603 {
1604   's :=
1605   #1 'nameptr :=
1606   ""
1607   s num.names$ 'numnames :=

```

```

1608 numnames 'namesleft :=
1609 { namesleft #0 > }
1610 { nameptr #1 >
1611   { " " * }
1612   'skip$
1613   if$
1614   s nameptr
1615   "{vv{ } }{ll{ }}{ f{ }}{ jj{ }}"
1616   format.name$ 't :=
1617   nameptr numnames = t "others" = and
1618   { "et al" * }
1619   { t sortify * }
1620   if$
1621   nameptr #1 + 'nameptr :=
1622   namesleft #1 - 'namesleft :=
1623 }
1624 while$
1625 }
1626

sort.format.title
1627 FUNCTION {sort.format.title}
1628 {
1629   't :=
1630   "A " #2
1631   "An " #3
1632   "The " #4 t chop.word
1633   chop.word
1634   chop.word
1635   sortify
1636   #1 global.max$ substring$
1637 }
1638

author.sort
1639 FUNCTION {author.sort}
1640 {
1641   author empty$
1642   { key empty$
1643     { "to sort, need author or key in " cite$ * warning$ "" }
1644     { key sortify }
1645     if$
1646   }
1647   { author sort.format.names }
1648   if$
1649 }
1650

author.title.sort
1651 FUNCTION {author.title.sort}
1652 {
1653   author empty$
1654   { title empty$
1655     { key empty$
1656       { "to sort, need author, title, or key in " cite$ * warning$ "" }
1657       { key sortify }
1658     }
1659     if$
1660     { title sort.format.title }
1661     if$
1662   }
1663   { author sort.format.names }
1664   if$
1665 }
1666

presort
1667 FUNCTION {presort}
1668 {

```

```

1669   author.title.sort
1670   "   "
1671   *
1672   year field.or.null sortify
1673   *
1674   "   "
1675   *
1676   title field.or.null
1677   sort.format.title
1678   *
1679   #1 entry.max$ substring$
1680   'sort.key$ :=
1681 }
1682
1683 ITERATE {presort}
1684
1685 SORT
1686
1687 </gost705s>
1688

```

1.12 Вычисление длины табуляции между номером и текстом

```

1689 STRINGS { longest.label }
1690
1691 INTEGERS { number.label longest.label.width }
1692

```

initialize.longest.label

```

1693 FUNCTION {initialize.longest.label}
1694 {
1695   "" 'longest.label :=
1696   #1 'number.label :=
1697   #0 'longest.label.width :=
1698 }
1699

```

longest.label.pass

```

1700 FUNCTION {longest.label.pass}
1701 {
1702   number.label int.to.str$ 'label :=
1703   number.label #1 + 'number.label :=
1704   label width$ longest.label.width >
1705   { label 'longest.label :=
1706     label width$ 'longest.label.width :=
1707   }
1708   'skip$
1709   if$
1710 }
1711
1712 EXECUTE {initialize.longest.label}
1713
1714 ITERATE {longest.label.pass}
1715

```

1.13 Формирование окружения thebibliography

1.13.1 Преамбула

begin.bib

```

1716 FUNCTION {begin.bib}
1717 {
1718   preamble$ empty$
1719   'skip$
1720   { preamble$ write$ newline$ }
1721   if$
1722   "\makeatletter\renewcommand{\@biblabel}[1]{#1.}\makeatother"
1723   write$ newline$
1724   "\ifx\undefined\BibPrefix\def\BibPrefix{}\else\fi"

```

```

1725 write$ newline$
1726 "\ifx\undefined\BibPostfix\def\BibPostfix{}\else\fi"
1727 write$ newline$
1728 "\ifx\undefined\BibEmph\def\BibEmph#1{#1}\else\fi"
1729 write$ newline$
1730 "\ifx\undefined\href\def\href#1#2{#2}\else\fi"
1731 write$ newline$
1732 "\ifx\undefined\url\def\url#1{\texttt{#1}}\else\fi"
1733 write$ newline$
1734 "\ifx\undefined\urlprefix\def\urlprefix{URL: }\else\fi"
1735 write$ newline$
1736 "\ifx\undefined\BibUrl\def\BibUrl#1{\urlprefix\url{#1}}\else\fi"
1737 write$ newline$
1738 "\ifx\undefined\BibUrlDate\long\def\BibUrlDate#1{({%"
1739 write$ newline$
1740 "\cyr\cyrd\cyra\cyrt\cyra\ %"
1741 write$ newline$
1742 "\cyro\cyrb\cyrr\cyra\cyrshch\cyre\cyrn\cyri\cyrya}: #1)}\else\fi"
1743 write$ newline$
1744 "\ifx\undefined\BibAnnote\long\def\BibAnnote#1{#1}\else\fi"
1745 write$ newline$
1746 "\begin{thebibliography}{\BibPrefix" longest.label * "\BibPostfix}" *
1747 write$ newline$
1748 "\def\selectlanguageifdefined#1{"
1749 write$ newline$
1750 "\expandafter\ifx\csname date#1\endcsname\relax"
1751 write$ newline$
1752 "\else\language\csname l@#1\endcsname\fi}"
1753 write$ newline$
1754 }
1755

```

1.13.2 Начало окружения

```

1756 EXECUTE {begin.bib}
1757
1758 EXECUTE {init.state.consts}
1759
1760 ITERATE {call.type$}
1761

```

1.13.3 Конец окружения

end.bib

```

1762 FUNCTION {end.bib}
1763 {
1764   newline$
1765   "\end{thebibliography}" write$ newline$
1766 }
1767
1768 EXECUTE {end.bib}
1769
1770 </gost705 | gost705s>

```

2 Предметный указатель

Функции

add.blank	4	format.names.rev	11
add.doi	20	format.names	11
and	4	format.number.series	15
article	20	format.number	17
author.after	18	format.pages.page	16
author.before	18	format.pages	16
author.sort	29	format.thesis.type	17
author.title.sort	29	format.title	12
bbl.cmplr	6	format.tr.number	17
bbl.docdisser	10	format.urldate	19
bbl.edition	6	format.url	19
bbl.etal	7	format.vol.num.pages	16
bbl.iin	9	format.volume	17
bbl.iissue	7	inbook	21
bbl.iiss	7	incollection	21
bbl.in	8	init.state.consts	2
bbl.mthesis	10	initialize.longest.label	30
bbl.nnr	8	inproceedings	22
bbl.nnumber	8	is.num	15
bbl.nr	8	longest.label.pass	30
bbl.number	8	manual	22
bbl.of	7	mastersthesis	23
bbl.pages	9	misc	23
bbl.page	9	multi.page.check	16
bbl.phdthesis	10	n.dashify	13
bbl.ppages	9	new.block.checka	5
bbl.ppage	9	new.block.checkb	5
bbl.techrep	10	new.block	3
bbl.vvolume	6	new.column	4
bbl.vvol	7	new.dblslash	3
begin.bib	30	new.semicolumn	4
bookauthor.after	18	new.sentence.checka	5
bookauthor.before	18	new.sentence.checkb	5
booklet	21	new.sentence	4
book	20	new.slash	3
change.language	2	non.stop	5
chop.word	28	not	4
conference	22	or	4
convert.edition	15	output.address.publisher.date	13
default.type	25	output.bibitem	13
docdisser	24	output.check	3
editor.organization.after	18	output.nonnull	2
either.or.check	14	output.url	19
emphasize	5	output	3
empty.misc.check	17	phdthesis	24
end.bib	31	presort	29
extract.num	15	proceedings	24
field.or.null	5	sort.format.names	28
fin.entry	3	sort.format.title	28
format.annotate	19	sortify	28
format.authors.after	12	space.word	6
format.authors	12	techreport	25
format.bookauthors.after	12	tie.connect	14
format.bookauthors	12	tie.or.space.connect	14
format.btitle	14	unpublished	25
format.bvolume	14	word.in	14
format.compiler.after	12		
format.date	13		
format.edition	15		
format.editors.after	12		
format.isbn	19		